# stichting mathematisch centrum

≥ MC

AFDELING TOURGEPASTE WISKUNDE (DEPARTMENT OF APPLIED MATHEMATICS)

TW 179/78

JUNI

N.M. TEMME

THE PARTY OF THE P

AN ALGORITHM WITH ALGOL 60 PROGRAM FOR THE COMPUTATION OF THE ZEROS OF ORDINARY BESSEL FUNCTIONS AND OF THEIR DERIVATIVES

Preprint

### 2e boerhaavestraat 49 amsterdam

Printed at the Mathematical Centre, 49, 2e Boerhaavestraat, Amsterdam.

The Mathematical Centre, founded the 11-th of February 1946, is a non-profit institution aiming at the promotion of pure mathematics and its applications. It is sponsored by the Netherlands Government through the Netherlands Organization for the Advancement of Pure Research (Z.W.O.).

An algorithm with ALGOL 60 program for the computation of the zeros of ordinary Bessel functions and of their derivatives

bу

N.M. Temme

### ABSTRACT

An algorithm is described for the computation of the zeros of the Bessel functions  $J_a(x)$ ,  $Y_a(x)$ ,  $J_a'(x)$  and  $Y_a'(x)$ . The algorithm is given as an ALGOL 60 procedure.

KEY WORDS & PHRASES: zeros of Bessel functions, ALGOL 60 procedure.

<sup>\*)</sup> This report will be submitted for publication elsewhere

#### 1. INTRODUCTION

In this note we describe an algorithm for the numerical computation of the zeros of the Bessel functions

(1.1) 
$$J_a(x), Y_a(x), J_a'(x), Y_a'(x).$$

When a is real, these functions each have an infinite number of real zeros, all of which are simple with the possible exception of x = 0. For non-negative a the s-th positive zeros of these functions are denoted by

(1.2) 
$$j_{a,s}$$
,  $y_{a,s}$ ,  $j'_{a,s}$ ,  $y'_{a,s}$  (s = 1,2,3,...)

except that x = 0 is counted as the first zero of  $J_0'(x)$ . For properties of the zeros the reader is referred to the literature, for instance to OLVER (1960), or to his contribution in ABRAMOWITZ & STEGUN (1964, Ch.9).

In section 2 we describe the algorithm for computing the zeros. It is based on a higher order Newton process. In section 3 we give methods for obtaining first approximations of the zeros. We use asymptotic expansions as given in the literature. In section 4 we give the implementation of the algorithm as an ALGOL 60 procedure. In order to compute values of the functions of (1.1) we use the ALGOL 60 procedures published earlier in TEMME (1976), which procedures are not incorporated here.

### ACKNOWLEDGEMENTS

I wish to thank Mr. Rob Montijn of the Mathematical Centre for his help in programming and testing the ALGOL 60 program.

### 2. HIGHER ORDER NEWTON PROCESS

It is well-known (cf., for instance, HOFSOMMER (1958)) that, if a function satisfies a second order differential equation, this fact may be used

with advantage in the computation of its zeros. In that event it is convenient to use a process in which derivatives of the function are needed. Such a process is the Newton-Raphson method. We use a higher order version of it.

Let f be the function, the roots of which are to be computed. Let  $\alpha$  be such a (simple!) root and let x be a first approximation. If the approximation is sufficiently accurate we have

(2.1) 
$$\alpha = x - \rho(x)[1 + \alpha_1(x)\rho(x) + \alpha_2(x)\rho^2(x) + \theta(\rho^3(x))]$$

with

$$\rho(x) = f(x)/f'(x)$$

(2.2) 
$$\alpha_{1}(x) = \frac{1}{2}f''(x)/f'(x)$$

$$\alpha_{2}(x) = \frac{1}{6}\{3[f''(x)/f'(x)]^{2} - f'''(x)/f'(x)\}.$$

By using the differential equation of the Bessel functions

(2.3) 
$$x^2 C_a''(x) + x C_a'(x) + (x^2 - a^2) C_a(x) = 0$$

the derivatives f'' and f''' can be eliminated. Furthermore we remark that derivatives of the Bessel functions can be expressed in terms of other Bessel functions by using

(2.4) 
$$C_a'(x) = \frac{a}{x} C_a(x) - C_{a+1}(x)$$
.

In (2.3) and (2.4)  $C_a$  stands for  $J_a$  or  $Y_a$ .

The expression (2.1) will be modified in several ways. A first modification comes from replacing  $\rho(x) = C_a(x)/C_a'(x)$  by  $r(x) = C_a(x)/C_{a+1}(x)$  (for the case of  $J_a$  and  $Y_a$ ) and from replacing  $\rho(x) = C_a'(x)/C_a''(x)$  by  $r(x) = C_a'(x)/C_a(x)$  (for  $J_a'$  and  $Y_a'$ ). In both cases  $\rho(x)/r(x) = O(1)$  for  $x \to \alpha$ , hence the 0-term remains of the same order, if we change  $\alpha_1$  and  $\alpha_2$  in the right way.

A second modification comes from writing the quadratic polynomial in

(2.1) as a Padé-fraction. An expression of the type

(2.5) 
$$f(x) = \alpha_0 + \alpha_1 x + \alpha_2 x^2 + O(x^3), \quad x \to 0,$$

may be converted by Padé-methods into

(2.6) 
$$f(x) = \frac{p_0 + p_1 x}{1 + q_1 x} + O(x^3), \quad x \to 0,$$

where, if  $\alpha_1 \neq 0$ ,

(2.7) 
$$p_0 = \alpha_0, \quad p_1 = (\alpha_1^2 - \alpha_0 \alpha_2)/\alpha_1, \quad q_1 = -\alpha_2/\alpha_1.$$

In an asymptotic sense (2.5) and (2.6) are equivalent, but numerically the fraction in (2.6) may yield a better approximation to f than the polynomial part in (2.5). From numerical experiments we concluded that this is indeed the case for the representation in (2.1).

A third modification of (2.1) comes from eliminating the derivatives of the Bessel functions appearing in  $\alpha_1$  and  $\alpha_2$  of (2.2). In the following subsections we give more details for the computation of the coefficients p and q appearing in the form obtained after the modifications, viz.

(2.8) 
$$\alpha = x - r(x) \frac{1 + pr(x)}{1 + qr(x)} + O(r^4(x)), \quad x \to \alpha.$$

### 2.1. The coefficients p and q for $J_a(x)$ and $Y_a(x)$

First we remark that

$$\rho(x) = C_{a}(x)/C'_{a}(x) = -r(x)/[1 - \frac{a}{x} r(x)],$$
  
$$r(x) = C_{a}(x)/C_{a+1}(x).$$

Further we have for  $\alpha_1$  and  $\alpha_2$  of (2.2) the relation

$$\alpha_{1}(x) = -\frac{1}{2x} [1 + \frac{a^{2} - x^{2}}{x} r(x) + O(r^{2}(x))], \quad x \to \alpha,$$

$$\alpha_{2}(x) = \frac{1 + x^{2} - a^{2}}{6x^{2}} + O(r(x)), \quad x \to \alpha.$$

This gives (2.8) (by using (2.5), (2.6), (2.7)), with

(2.9) 
$$p = \frac{1 + 4x^2 - 4a^2}{6x(2a+1)}, \qquad q = \frac{2x^2 - 1 - 6a - 8a^2}{3x(2a+1)}.$$

Formula (2.8) is then used in the sense that if x is an approximation for the s-th zero of  $J_a(x)$  or  $Y_a(x)$ , then

(2.10) 
$$j_{a,s}, y_{a,s} = x - r(x) \frac{1 + pr(x)}{1 + qr(x)}$$

is a better approximation, with p and q given in (2.9) and

$$r(x) = J_a(x)/J_{a+1}(x)$$
 or  $r(x) = Y_a(x)/Y_{a+1}(x)$ .

## 2.2. The coefficients p and q for $J'_a(x)$ and $Y'_a(x)$

In this case we write in (2.1)

$$\rho(x) = \frac{C_a'(x)}{C_a''(x)} = r(x) \frac{x^2}{a^2 - x^2 - xr(x)}, \quad r(x) = C_a'(x)/C_a(x)$$

and we change  $\alpha_{1}$  and  $\alpha_{2}$  as before. The result is

(2.11) 
$$j'_{a,s}, y'_{a,s} = x - \frac{x^2}{a^2 - x^2} r(x) \frac{1 + pr(x)}{1 + qr(x)}$$

where x is an approximation of the s-th zero of  $J_a'(x)$  or  $Y_a'(x)$ ,

$$r(x) = J_a'(x)/J_a(x)$$
 or  $r(x) = Y_a'(x)/Y_a(x)$ 

and

$$p = \frac{4x(a^2-x^2)}{6(a^2+x^2)} \left[1 + \frac{10x^2a^2 + 3x^4 - a^4}{4(a^2-x^2)^3}\right],$$

$$q = \frac{4x(a^2-x^2)}{6(a^2+x^2)} \left[1 + \frac{8x^2a^2 + 3x^4 + a^4}{2(a^2-x^2)^3}\right].$$

Formulas (2.10) and (2.11) will be repeatedly used until the approximation is sufficiently accurate. In the next section we describe methods for obtaining first (i.e., initial) approximations to the zeros.

#### 3. FIRST APPROXIMATIONS TO THE ZEROS

From ABRAMOWITZ & STEGUN (1964, p.371) we take the following expansions of the s-th zero  $j_{a,s}$  (or  $y_{a,s}$ ) of  $J_a(x)$  (or  $Y_a(x)$ )

(3.1) 
$$j_{a,s}, y_{a,s} = \beta - \frac{\mu-1}{8\beta} \left[ 1 + \frac{4(7\mu-31)}{3(8\beta)^2} + \frac{32(83\mu^2-982\mu+3779)}{15(8\beta)^4} \right] + O(\beta^{-7}),$$

for s  $\rightarrow \infty$ , where  $\mu = 4a^2$  and

$$\beta = (s + \frac{1}{2}a - \frac{1}{4})\pi \quad \text{for } j_{a,s}$$

$$\beta = (s + \frac{1}{2}a - \frac{3}{4})\pi \quad \text{for } y_{a,s}.$$

This approximation can be written as (following the method described by (2.5), (2.6), (2.7))

(3.3) 
$$j_{a,s}, y_{a,s} = \beta - \frac{\mu - 1}{8\beta} \left[ \frac{1 - p/(8\beta)^2}{1 - q/(8\beta)^2} \right] + O(\beta^{-7}),$$

for  $s \rightarrow \infty$ , with  $\beta$  given in (3.2) and

$$p = \frac{4(253\mu^2 - 3722\mu + 17869)}{15(7\mu - 31)}, \quad q = \frac{8(83\mu^2 - 982\mu + 3779)}{5(7\mu - 31)}.$$

The relations in (3.1) and (3.3) are valid for  $s \to \infty$ . However, for small values of a they give good approximations for small values of s, even for s = 1. For a = 0, s = 1, (3.3) gives  $j_{0,1} = 2.4052...$  while in 7 significant digits it is 2.404825. Hence, the absolute error is 0.00043. (Formula (3.1) gives a result with absolute error 0.0016.) For large values of a (3.3) (and (3.1)) is useless for the smaller s-values. Other approximations will be given for this case. First we give the results for the zeros of  $J_a'(x)$  and  $Y_a'(x)$  for small a.

In this case we have for large s

(3.4) 
$$j_{a,s}, y_{a,s}' = \beta' - \frac{\mu+3}{8\beta'} - \frac{4(7\mu^2 + 82\mu - 9)}{3(8\beta')^3} - \frac{32(83\mu^3 + 2075\mu^2 - 3039\mu + 3537)}{15(8\beta')^5} + O(\beta'^{-7}),$$

where  $\mu = 4a^2$  and

$$\beta' = (s + \frac{1}{2}a - \frac{3}{4})\pi \quad \text{for j'}_{a,s}$$

$$\beta' = (s + \frac{1}{2}a - \frac{1}{4})\pi \quad \text{for y'}_{a,s}.$$

The Padé-version of (3.4) reads as follows:

(3.6) 
$$j'_{a,s}, y'_{a,s} = \beta' - \frac{1}{8\beta'} \left[ \frac{p_0 - p_1/(8\beta')^2}{1 - q_1/(8\beta')^2} \right] + O(\beta'^{-7}),$$

where

$$\begin{aligned} \mathbf{p}_0 &= \mu + 3, \\ \mathbf{p}_1 &= \frac{4(253\mu^4 + 8204\mu^3 - 13874\mu^2 - 26100\mu + 63261)}{15(7\mu^2 + 82\mu - 9)} \\ \mathbf{q}_1 &= \frac{8(83\mu^3 + 2075\mu^2 - 3039\mu + 3537)}{5(7\mu^2 + 82\mu - 9)}. \end{aligned}$$

As mentioned earlier, the approximations (3.3) and (3.6) are valid for small values of a. If a is large (this "large" will be specified in more detail further on) we need other approximations for the early zeros. These can be obtained from Olver's result on uniform asymptotic expansions for the Bessel functions. We use the formulas in ABRAMOWITZ & STEGUN (1964, p.371, 9.5.22 through 9.5.26). The zeros are expressed in terms of the zeros of Airy functions. The formulas are valid for all  $s \ge 1$ .

For  $j_{a,s}$ ,  $j'_{a,s}$  the formulas read as follows:

$$j_{a,s} = az(\zeta) + \frac{f_1(\zeta)}{a} + O(a^{-2})$$

$$j'_{a,s} = az(\zeta) + \frac{g_1(\zeta)}{a} + O(a^{-2})$$

$$a \to \infty$$

with

(3.8) 
$$\zeta = a^{-2/3} a_s \quad \text{for } j_{a,s}$$
$$\zeta = a^{-2/3} a_s' \quad \text{for } j_{a,s}'$$

(where  $a_s, a_s'$  is the s-th negative zero of  $A_i(z)$ ,  $A_i'(z)$ ) and with  $z(\zeta)$  defined implicitly by

(3.9) 
$$\frac{2}{3}(-\zeta)^{3/2} = (z^2-1)^{\frac{1}{2}} - \arccos(1/z), \quad \zeta \le 0, \quad z \ge 1.$$

The functions  $f_1$  and  $g_1$  appearing in (3.7) are given by

$$f_{1}(\zeta) = -z(\zeta)\zeta^{-1}h(\zeta)\left[5\zeta^{-1}/48 + h(\zeta)\left(\frac{5}{z^{2}-1} + 3\right)/24\right],$$

$$g_{1}(\zeta) = z(\zeta)\zeta^{-1}h(\zeta)\left[7\zeta^{-1}/48 + h(\zeta)\left(\frac{7}{z^{2}-1} + 9\right)/24\right],$$

$$h(\zeta) = \left[\zeta(1-z^{2})\right]^{\frac{1}{2}}.$$

The expansions for  $y_{a,s}$ ,  $y'_{a,s}$  are given by (3.7) if in (3.8) the zeros  $b_s$ ,  $b'_s$  of  $B_i(z)$ ,  $B'_i(z)$  are used instead of  $a_s$ ,  $a'_s$ .

From numerical experiments it follows that (3.7) gives good initial approximations, even if a is rather small. For a = 3 and s = 1 the approximations based on (3.7) are of the same order of accuracy as those based on (3.3) or (3.6). However, the numerical process based on (3.7) is more intricate than the other one, since we need the inversion of (3.9), which must be carried out numerically. At the end of this section we pay attention to this problem, first, however, we give an indication for which values of a and s the approximations in (3.3), (3.6) and (3.7) are to be used.

A safe bound for using (3.3) and (3.6) is s > 3a, even when a is large. In order to take into account the smaller values of s and a we propose the  ${\tt Following}$  criterion: if

$$s \ge 3a - 8$$
,  $a \ge 0$ 

then (3.3) and (3.6) are to be used, otherwise (3.7). Consequently, for  $0 \le a \le 3$ , we use (3.3) and (3.6) for all s; if a becomes larger than 3 it is better to use (3.7) for the early zeros.

Finally we describe a method for the inversion of (3.9). By substituting

 $z = 1/\cos(x)$ ,  $y = \frac{2}{3}(-\zeta)^{3/2}$ , it follows that we need the inversion of (3.10) tg x - x = y,  $y \ge 0$ ,  $0 \le x < \frac{\pi}{2}$ .

Since we need the solution of this equation for a first approximation in (3.7), it is not necessary to solve (3.10) with high accuracy precision. Four significant digits in x is enough.

For small y we can expand

(3.11) 
$$x = \sum_{i=0}^{\infty} c_{2i+1} p^{2i+1}, \quad p = (3y)^{1/3},$$

where the first few coefficients are

$$c_1 = 1$$
,  $c_3 = -2/15$ ,  $c_5 = 3/175$ ,  $c_7 = -2/1575$ .

For large y we write  $x = \frac{\pi}{2} - \xi$ , and (3.10) becomes

tg 
$$\xi = \frac{p}{1 - p\xi}$$
,  $p = 1/(y + \frac{\pi}{2})$ ,

and again we can expand

(3.12) 
$$\xi = \sum_{i=0}^{\infty} d_{2i+1} p^{2i+1}.$$

In this case the first coefficients are

$$d_1 = 1$$
,  $d_3 = 2/3$ ,  $d_5 = 13/15$ ,  $d_7 = 146/105$ ,  $d_9 = 781/315$ ,  $d_{11} = 16328/3465$ .

If we use the above given coefficients  $c_i$  and  $d_i$ , both (3.11) and (3.12) give about four correct significant digits for y = 1.

### 4. AN ALGOL 60 PROCEDURE

The heading of the procedure given in this section reads as follows:

procedure bess zeros (a,n,z,d,e); value a,n,d,e;

real a,e; integer n,d; array z;

he meaning of the formal parameters is:

- :: <arithmetic expression>;
  the order of the Bessel function, a ≥ 0.
- : <arithmetic expression>;
  the number of zeros to be computed, n ≥ 1.
- : <array identifier>;
   array z[1:n];
   exit: z[j] is the j-th zero of the selected Bessel function.
- : <arithmetic expression>;
   the choice of d determines the type of the Bessel function of which
   the zeros are to be computed:

if d = 1 then  $J_a$ , if d = 2 then  $Y_a$ , if d = 3 then  $J_a$ , if d = 4 then  $Y_a$ .

: <arithmetic expression>;
 the desired relative accuracy in the zeros;
 e should be larger than the machine accuracy.

The procedure calls for the nonlocal procedure besspqa, which is ublished in TEMME (1976). It computes the functions  $P_a$ ,  $Q_a$  appearing in he relations

$$J_{a}(x) = \left[2/(\pi x)\right]^{\frac{1}{2}} \left[P_{a}(x) \cos \chi - Q_{a}(x) \sin \chi\right],$$

$$Y_a(x) = [2/(\pi x)]^{\frac{1}{2}} [P_a(x) \sin x + Q_a(x) \cos x],$$

ith  $\chi = x - \pi(2a+1)/4$ , and  $P_{a+1}$ ,  $Q_{a+1}$  which are defined equivalently. If for the computation of a zero more than 5 iterations are needed in

ne Newton process then the last computed value of the zero is accepted.

therwise the computation is stopped if consecutive approximations agree

within a relative precision e (see the heading of the procedure) with each other. From various tests it followed that not more than 3 iterations are needed if  $e = 10^{-1}$ 3. In many cases only 1 iteration is needed for obtaining this precision.

```
procedure bess zeros(a,n,z,d,e); value a,n,d,e;
            real a,e; integer n,d; array z;
comment computes z[1],...z[n], the first n zeros of a bessel function.
   the choice of d determines the type of the bessel function :
   if d=1 then ja
                        else
   if d=2 then ya
                        else
   if d=3 then ja-prime else
   if d=4 then ya-prime.
   a is the order of the bessel function, it must be non-negative.
   e is a measure for the relative accuracy;
       real aa,a2,b,bb,c,chi,co,mu,mu2,mu3,mu4,p,pi,pa,pa1,p0,p1,pp1,
              q,qa,qa1,q1,qq1,ro,si,t,tt,u,v,w,x,xx,x4,y; <u>integer</u> j,s;
           procedure fi(y); value y; real y;
    comment computes fi from the equation
          tan(fi)-fi=y, where y>=0.
    the relative accuracy is at least 5 digits;
    <u>if</u> y=0
              then fi:=0
    <u>if</u> y>"5 <u>then</u> fi:=1.570796
                                    else
    begin real r,p,pp;
        if y<1 then
        <u>begin</u> p:=(3*y)**(1/3); pp:=p*p;
           p:=p*(1+pp*(-210+pp*(27-2*pp))/1575)
            <u>else</u>
        begin p:=1/(y+1.570796); pp:=p*p;
            p := 1.570796 - p*(1+pp*(2310+pp*(3003+pp*(4818+pp*
                    (8591+pp*16328))))/3465)
        end;
        pp:=(y+p)*(y+p); r:=(p-arctan(p+y))/pp;
        fi:=p-(1+pp)*r*(1+r/(p+y))
    end fi;
    <u>real</u>
           procedure r;
    begin besspqa(a,x,e,pa,qa,pa1,qa1);
        chi:=x-pi*(a/2+0.25);
        si :=sin(chi); co:=cos(chi);
        r:= \underline{if} d=1 \underline{then} (pa*co-qa*si)/(pa1*si+qa1*co) \underline{else}
             if d=2 then (pa*si+qa*co)/(qa1*si-pa1*co) else
             if d=3 then a/x-(pa1*si+qa1*co)/(pa*co-qa*si)
                             a/x-(qa1*si-pa1*co)/(pa*si+qa*co)
    end r;
```

```
pi:=4*arctan(1); aa:=a*a; mu:=4*aa; mu2:=mu*mu;
    mu3:=mu*mu2: mu4:=mu2*mu2:
    if d<3 then
    begin p:=7*mu-31; p0:=mu-1;
        p1:=4*(253*mu2-3722*mu+17869)/15/p*p0:
        a1:=8*( 83*mu2- 982*mu+ 3779)/ 5/p
    end
          <u>else</u>
    begin p:=7*mu2+82*mu-9; p0:=mu+3;
        p1: = (4048 * mu4 + 131264 * mu3 - 221984 * mu2 - 417600 * mu + 1012176)/60/p;
        q1:=1.6*(83*mu3+2075*mu2-3039*mu+3537)/p
    end :
    t:= <u>if</u> d=1 <u>or</u> d=4 <u>then</u> 0.25 <u>else</u> 0.75; tt:=4*t;
    if d<3 then
    begin pp1:= 5/48; qq1:= -5/36
                                         <u>end</u>
                                                <u>else</u>
    begin pp1:=-7/48; qq1:= 35/288 end;
    y:= 3*pi/8; bb:= a**(-2/3);
    for s:=1 step 1 until n do
             if a=0 and s=1 and d=3 then
    begin
        begin x:=0; j:=0 end
                                    else
        begin
                <u>if</u> s >= 3*a -8
                                     then
             begin b:=(s+a/2-t)*pi; c:=1/b/b/64;
                 x := b-1/b/8*(p0-p1*e)/(1-q1*e)
                    else
             end
             <u>begin</u>
                    <u>if</u> s=1 <u>then</u>
                 begin x:= if d=1
                                         <u>then</u> -2.33811
                                                           <u>else</u>
                               <u>if</u>
                                   d=2
                                        <u>then</u> -1.17371
                                                           e<u>lse</u>
                                        t<u>hen</u> -1.01879
                                                          else -2.29444
                               <u>if</u>
                                   d=3
                 end
                        <u>else</u>
                 <u>begin</u> x:= y*(4*s-tt); v:= 1/x/x;
                         x := -x^{**}(2/3)^{*}(1+v^{*}(pp1+qq1^{*}v))
                 end;
                 u:=x*bb; v:=fi(2/3*(-u)**1.5);
                 w:=1/\cos(v); xx:=1-w*w; c:=sqrt(u/xx);
                 x:=w*(a+c/a/u*
                  (if d<3 then -5/48/u-c*(-5/24/xx+1/8)
                                    7/48/u+c*(-7/24/xx+3/8))
                             else
             end; j:=0;
             l1: xx:=x*x; x4:=xx*xx; a2:=aa-xx; ro:=r; j:=j+1;
                 if d<3 then
                 begin u:=ro; p:=(1-4*a2)/6/x/(2*a+1);
                      q := (2*(xx-mu)-1-6*a)/3/x/(2*a+1)
                       <u>else</u>
                  end
                 <u>begin</u> u:=-xx*ro/a2; v:=2*x*a2/(aa+xx)/3;
                     w:=a2*a2*a2;
                      q:=v*(1+(mu2+32*mu*xx+48*x4)/32/w);
                      p:=v*(1+(-mu2+40*mu*xx+48*x4)/64/w)
             w:=u^*(1+p^*ro)/(1+q^*ro); x:=x+w;
             if abs(w/x) > e and j < 5 then goto 11
         end ; z[s]:=x
    end
end bess zeros;
```

### REFERENCES

- ABRAMOWITZ, M. & I.A. STEGUN (eds.), (1964), Handbook of mathematical functions with formulas, graphs and mathematical tables,

  Appl. Math. Ser. 55, U.S. Govt. Printing Office, Washington, D.C.
- HOFSOMMER, D.J. (1958), Note on the computation of the zeros of functions satisfying a second order differential equation, Math. Tables
  Aid Comput. XII, pp.58-60.
- OLVER, F.W.J. (ed.), (1960), Royal Society Mathematical Tables, vol. 7,

  Bessel functions, Part III: Zeros and associated values,

  Cambridge Univ. Press, Cambridge.
- TEMME, N.M. (1976), On the numerical evaluation of the ordinary Bessel function of the second kind, J. Comp. Physics, vol. 21, pp.343-350.